
Objects



Plan:

Plan:

1. Identify some characteristic “core features” of object-oriented
programming

2. Develop two different analyses of these features:

2.1 A translation into a lower-level language
2.2 A direct, high-level formalization of a simple object-oriented

language (“Featherweight Java”)



The Translational Analysis

Our first goal will be to show how many of the basic features of
object-oriented languages

dynamic dispatch
encapsulation of state
inheritance
this
super

can be understood as “derived forms” in a lower-level language
with a rich collection of primitive features:

(higher-order) functions
records
references
recursion
subtyping



The Translational Analysis

For simple objects and classes, this translational analysis works
very well.

When we come to more complex features (in particular, classes
with this), it becomes less satisfactory, leading us to the more
direct treatment in the following chapter.



Concepts



The Essence of Objects

What “is” object-oriented programming?

The term is used widely, but there are some core features that are
strongly implied by it.

Dynamic dispatch

Perhaps the most basic characteristic of object-oriented
programming is dynamic dispatch: when an operation is invoked on
an object, the ensuing behavior depends on the object itself, rather
than being fixed (as when we apply a function to an argument).

Two objects of the same type (i.e., responding to the same set of
operations) may be implemented internally in completely different
ways.

This is late binding for function calls.



Example (in Java)

class A {
int x = 0;
int m() { x = x+1; return x; }
int n() { x = x-1; return x; }

}

class B extends A {
int m() { x = x+5; return x; }

}

class C extends A {
int m() { x = x-10; return x; }

}

Note that (new B()).m() and (new C()).m() invoke completely
different code!



Aside: Late Binding

Object-oriented programming started developing during the 60’s
and 70’s. Alan Kay claims to have used late binding as a driving
concept in exploring object-oriented design.

The idea is, let’s be honest: we do not know how to program, nor
to write programming languages. Thus, let us defer as many
decisions as possible. Let us bind many things as late as possible.

The clearest application of this is dynamic dispatch of methods.



Encapsulation

In most OO languages, each object consists of some internal state
encapsulated with a collection of method implementations
operating on that state.

I state directly accessible to methods

I state inaccessible from outside the object



Encapsulation

In Java, encapsulation of internal state is optional. For full
encapsulation, fields must be marked protected:

class A {
protected int x = 0;
int m() { x = x+1; return x; }
int n() { x = x-1; return x; }

}

class B extends A {
int m() { x = x+5; return x; }

}

class C extends A {
int m() { x = x-10; return x; }

}

The code (new B()).x is not allowed.



Side note: Objects vs. ADTs

The encapsulation of state with methods offered by objects is a
form of information hiding.

A somewhat different form of information hiding is embodied in
the notion of an abstract data type (ADT).



Side note: Objects vs. ADTs

An ADT comprises:

I A hidden representation type X

I A collection of operations for creating and manipulating
elements of type X.

Similar to OO encapsulation in that only the operations provided
by the ADT are allowed to directly manipulate elements of the
abstract type.

But different in that there is just one (hidden) representation type
and just one implementation of the operations — no dynamic
dispatch.

Both styles have advantages.

Caveat: In the OO community, the term “abstract data type” is
often used as more or less a synonym for “object type.” This is
unfortunate, since it confuses two rather different concepts.



Subtyping and Encapsulation

The “type” (or “interface” in Smalltalk terminology) of an object
is just the set of operations that can be performed on it (and the
types of their parameters and results); it does not include the
internal representation.

Object interfaces fit naturally into a subtype relation.

An interface listing more operations is “better” than one
listing fewer operations.

This gives rise to a natural and useful form of polymorphism: we
can write one piece of code that operates uniformly on any object
whose interface is “at least as good as I” (i.e., any object that
supports at least the operations in I).



Example

// ... class A and subclasses B and C as above...

class D {
int p (A myA) { return myA.m(); }

}

...

D d = new D();
int z = d.p (new B());
int w = d.p (new C());



Inheritance

Objects that share parts of their interfaces will typically (though
not always) share parts of their behaviors.

To avoid duplication of code, want to write the implementations of
these behaviors in just one place.
=⇒ inheritance



Inheritance

Basic mechanism of inheritance: classes

A class is a data structure that can be

I instantiated to create new objects (“instances”)

I refined to create new classes (“subclasses”)

N.b.: some OO languages offer an alternative mechanism, called
delegation, which allows new objects to be derived by refining the
behavior of existing objects.



Example

class A {
protected int x = 0;
int m() { x = x+1; return x; }
int n() { x = x-1; return x; }

}

class B extends A {
int o() { x = x*10; return x; }

}

An instance of B has methods m, n, and o. The first two are
inherited from A.



Ubiquitous this

OO languages provide ubiquitous access to this, the current
method receiver.

This is a form of open recursion, and it is late binding of the
receiver of a method.

The interesting thing is that this might be an instance of a
subclass, not the class you are currently looking at! So, the system
must be sure to allow this’s actual class at run time to override
the definitions of the current class.



Examples

class E {
protected int x = 0;
int m() { x = x+1; return x; }
int n() { x = x-1; return this.m(); }

}

class F extends E {
int m() { x = x+100; return x; }

}

Quick check:

I What does (new E()).n() return?

I What does (new F()).n() return?



Calling “super”

It is sometimes convenient to “re-use” the functionality of an
overridden method.

Java provides a mechanism called super for this purpose.



Example

class E {
protected int x = 0;
int m() { x = x+1; return x; }
int n() { x = x-1; return this.m(); }

}

class G extends E {
int m() { x = x+100; return super.m(); }

}

What does (new G()).n() return?



Getting down to details
(in the lambda-calculus)...



Simple objects with encapsulated state

class Counter {
protected int x = 1; // Hidden state
int get() { return x; }
void inc() { x++; }

}

void inc3(Counter c) {
c.inc(); c.inc(); c.inc();

}

Counter c = new Counter();
inc3(c);
inc3(c);
c.get();

How do we encode objects in the lambda-calculus?



Objects

c = let x = ref 1 in
{get = λ_:Unit. !x,
inc = λ_:Unit. x:=succ(!x)};

=⇒ c : Counter
where
Counter = {get:Unit→Nat, inc:Unit→Unit}



Objects

inc3 = λc:Counter. (c.inc unit; c.inc unit; c.inc unit);
=⇒ inc3 : Counter → Unit

(inc3 c; inc3 c; c.get unit);
=⇒ 7



Object Generators

newCounter =
λ_:Unit. let x = ref 1 in

{get = λ_:Unit. !x,
inc = λ_:Unit. x:=succ(!x)};

=⇒ newCounter : Unit → Counter



Grouping Instance Variables

Rather than a single reference cell, the states of most objects
consist of a number of instance variables or fields.

It will be convenient (later) to group these into a single record.

newCounter =
λ_:Unit. let r = {x=ref 1} in

{get = λ_:Unit. !(r.x),
inc = λ_:Unit. r.x:=succ(!(r.x))};

The local variable r has type CounterRep = {x: Ref Nat}



Subtyping and Inheritance

class Counter {
protected int x = 1;
int get() { return x; }
void inc() { x++; }

}

class ResetCounter extends Counter {
void reset() { x = 1; }

}

ResetCounter rc = new ResetCounter();
inc3(rc);
rc.reset();
inc3(rc);
rc.get();



Subtyping

ResetCounter = {get:Unit→Nat,
inc:Unit→Unit,
reset:Unit→Unit};

newResetCounter =
λ_:Unit. let r = {x = ref 1} in

{get = λ_:Unit. !(r.x),
inc = λ_:Unit. r.x:=succ(!(r.x)),
reset = λ_:Unit. r.x:=1};

=⇒ newResetCounter : Unit → ResetCounter



Subtyping

rc = newResetCounter unit;
(inc3 rc; rc.reset unit; inc3 rc; rc.get unit);
=⇒ 4



Simple Classes

The definitions of newCounter and newResetCounter are
identical except for the reset method.

This violates a basic principle of software engineering:

Each piece of behavior should be implemented in just one
place in the code.



Reusing Methods

Idea: could we just re-use the methods of some existing object to
build a new object?

resetCounterFromCounter =
λc:Counter. let r = {x = ref 1} in

{get = c.get,
inc = c.inc,
reset = λ_:Unit. r.x:=1};

No: This doesn’t work properly because the reset method does
not have access to the local variable r of the original counter.

=⇒ classes

Classes

A class is a run-time data structure that can be

1. instantiated to yield new objects

2. extended to yield new classes



Classes

To avoid the problem we observed before, what we need to do is to
separate the definition of the methods

counterClass =
λr:CounterRep.
{get = λ_:Unit. !(r.x),
inc = λ_:Unit. r.x:=succ(!(r.x))};

=⇒ counterClass : CounterRep → Counter

from the act of binding these methods to a particular set of
instance variables:

newCounter =
λ_:Unit. let r = {x=ref 1} in

counterClass r;
=⇒ newCounter : Unit → Counter



Defining a Subclass

resetCounterClass =
λr:CounterRep.
let super = counterClass r in

{get = super.get,
inc = super.inc,
reset = λ_:Unit. r.x:=1};

=⇒ resetCounterClass : CounterRep → ResetCounter

newResetCounter =
λ_:Unit. let r = {x=ref 1} in resetCounterClass r;

=⇒ newResetCounter : Unit → ResetCounter



Overriding and adding instance variables

class Counter {
protected int x = 1;
int get() { return x; }
void inc() { x++; }

}

class ResetCounter extends Counter {
void reset() { x = 1; }

}

class BackupCounter extends ResetCounter {
protected int b = 1;
void backup() { b = x; }
void reset() { x = b; }

}



Adding instance variables

In general, when we define a subclass we will want to add new
instances variables to its representation.

BackupCounter = {get:Unit→Nat, inc:Unit→Unit,
reset:Unit→Unit, backup: Unit→Unit};

BackupCounterRep = {x: Ref Nat, b: Ref Nat};

backupCounterClass =
λr:BackupCounterRep.
let super = resetCounterClass r in

{get = super.get,
inc = super.inc,
reset = λ_:Unit. r.x:=!(r.b),
backup = λ_:Unit. r.b:=!(r.x)};

=⇒
backupCounterClass : BackupCounterRep → BackupCounter



Notes:

I backupCounterClass both extends (with backup) and
overrides (with a new reset) the definition of counterClass

I subtyping is essential here (in the definition of super)

backupCounterClass =
λr:BackupCounterRep.
let super = resetCounterClass r in

{get = super.get,
inc = super.inc,
reset = λ_:Unit. r.x:=!(r.b),
backup = λ_:Unit. r.b:=!(r.x)};



Calling super

Suppose (for the sake of the example) that we wanted every call to
inc to first back up the current state. We can avoid copying the
code for backup by making inc use the backup and inc methods
from super.

funnyBackupCounterClass =
λr:BackupCounterRep.
let super = backupCounterClass r in

{get = super.get,
inc = λ_:Unit. (super.backup unit; super.inc unit),
reset = super.reset,
backup = super.backup};

=⇒
funnyBackupCounterClass : BackupCounterRep → BackupCounter



Calling between methods

What if counters have set, get, and inc methods:

SetCounter = {get:Unit→Nat, set:Nat→Unit, inc:Unit→Unit};

setCounterClass =
λr:CounterRep.

{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. r.x:=(succ r.x) });

Bad style: The functionality of inc could be expressed in terms of
the functionality of get and set.

Can we rewrite this class so that the get/set functionality appears
just once?

Calling between methods

In Java we would write:
class SetCounter {

protected int x = 0;
int get () { return x; }
void set (int i) { x = i; }
void inc () { this.set( this.get() + 1 ); }

}



Better...

setCounterClass =
λr:CounterRep.
fix
(λthis: SetCounter.
{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. this.set (succ (this.get unit))});

Check: the type of the inner λ-abstraction is
SetCounter→SetCounter, so the type of the fix expression is
SetCounter.

This is just a definition of a group of mutually recursive functions.



Note that the fixed point in

setCounterClass =
λr:CounterRep.
fix
(λthis: SetCounter.
{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. this.set (succ (this.get unit))});

is “closed” — we “tie the knot” when we build the record.

So this does not model the behavior of this (or self) in real OO
languages.



Idea: move the application of fix from the class definition...

setCounterClass =
λr:CounterRep.

λthis: SetCounter.
{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. this.set (succ(this.get unit))};

...to the object creation function:

newSetCounter =
λ_:Unit. let r = {x=ref 1} in

fix (setCounterClass r);

In essence, we are switching the order of fix and
λr:CounterRep...



Note that we have changed the types of classes from...
setCounterClass =

λr:CounterRep.
fix
(λthis: SetCounter.
{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. this.set (succ (this.get unit))});

=⇒ setCounterClass : CounterRep → SetCounter

... to:
setCounterClass =

λr:CounterRep.
λthis: SetCounter.
{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. this.set (succ(this.get unit))};

=⇒
setCounterClass : CounterRep → SetCounter → SetCounter



Using this

Let’s continue the example by defining a new class of counter
objects (a subclass of set-counters) that keeps a record of the
number of times the set method has ever been called.

InstrCounter = {get:Unit→Nat, set:Nat→Unit,
inc:Unit→Unit, accesses:Unit→Nat};

InstrCounterRep = {x: Ref Nat, a: Ref Nat};



instrCounterClass =
λr:InstrCounterRep.

λthis: InstrCounter.
let super = setCounterClass r this in
{get = super.get,
set = λi:Nat. (r.a:=succ(!(r.a)); super.set i),
inc = super.inc,
accesses = λ_:Unit. !(r.a)};

=⇒ instrCounterClass :

InstrCounterRep → InstrCounter → InstrCounter

Notes:

I the methods use both this (which is passed as a parameter)
and super (which is constructed using this and the instance
variables)

I the inc in super will call the set defined here, which calls
the superclass set

I suptyping plays a crucial role (twice) in the call to
setCounterClass



One more refinement...



A small fly in the ointment

The implementation we have given for instrumented counters is
not very useful because calling the object creation function

newInstrCounter =
λ_:Unit. let r = {x=ref 1, a=ref 0} in

fix (instrCounterClass r);
will cause the evaluator to diverge!

Intuitively (see TAPL for details), the problem is the “unprotected”
use of this in the call to setCounterClass in
instrCounterClass:

instrCounterClass =
λr:InstrCounterRep.

λthis: InstrCounter.
let super = setCounterClass r this in

...



To see why this diverges, consider a simpler example:
ff = λf:Nat→Nat.

let f′ = f in
λn:Nat. 0

=⇒ ff : (Nat→Nat) → (Nat→Nat)

Now:
fix ff −→ let f′ = (fix ff) in λn:Nat. 0

−→ let f′ = ff (fix ff) in λn:Nat. 0
−→ uh oh...



One possible solution

Idea: “delay” this by putting a dummy abstraction in front of it...

setCounterClass =
λr:CounterRep.
λthis: Unit→SetCounter.

λ_:Unit.
{get = λ_:Unit. !(r.x),
set = λi:Nat. r.x:=i,
inc = λ_:Unit. (this unit).set

(succ((this unit).get unit))};
=⇒ setCounterClass :

CounterRep → (Unit→SetCounter) → (Unit→SetCounter)

newSetCounter =
λ_:Unit. let r = {x=ref 1} in

fix (setCounterClass r) unit;



Similarly:

instrCounterClass =
λr:InstrCounterRep.
λthis: Unit→InstrCounter.

λ_:Unit.
let super = setCounterClass r this unit in

{get = super.get,
set = λi:Nat. (r.a:=succ(!(r.a)); super.set i),
inc = super.inc,
accesses = λ_:Unit. !(r.a)};

newInstrCounter =
λ_:Unit. let r = {x=ref 1, a=ref 0} in

fix (instrCounterClass r) unit;



Success

This works, in the sense that we can now instantiate
instrCounterClass (without diverging!), and its instances
behave in the way we intended.

However, all the “delaying” we added has an unfortunate side
effect: instead of computing the “method table” just once, when
an object is created, we will now re-compute it every time we
invoke a method!

Section 18.12 in TAPL shows how this can be repaired by using
references instead of fix to “tie the knot” in the method table.

Recap

We implemented object-oriented features on top of
function-language features:

I Dynamic dispatch: use records of functions.

I Encapsulation: use variable capture, so that these functions
see the hidden state but the callers of the functions do not.

I Subtyping: use record subtyping, and get it for free.

I Inheritance: introduce classes, and separate out the
representation records.

I Ubiquitous this: tricky! Bind this via fix. . . but do not call
fix in the class definition. . . and do fix on a delayed
Unit→Object function instead of directly on the Object.

